

GDL EDITOR

SAMPLES (v0.93)

FUNDED BY: **Cambio⁺** Healthcare Systems (<http://www.cambio.se>)

BODY MASS INDEX CALCULATION

This example will describe how to create a simple guideline to calculate the body mass index using the formula:

$$\text{BMI} = \frac{\text{mass}(\text{kg})}{(\text{height}(\text{m}))^2}$$

The description of the guide should include all the necessary information to make sure it is used in the correct context.

To add a new rule, we click on the *Add rule* button. We include the name of the rule and hit *Accept*.




Figure 1: Creating a new rule

After this step is performed, we will begin the rule editing. We should make sure that the elements needed have the correct units. We want to add two conditions: weight is measured in kilograms and height in centimeters. Double clicking twice on the *Compare (Attribute)* condition, we will add two empty conditions to the rule.




Figure 2: Adding conditions to specify units

Now we can specify the units for both elements we are going to obtain from the EHR (weight and height). Clicking on the *Element attribute* link, we can select the attribute we are going to use. Since we don't have yet any archetypes defined on the guide, we will have to click on the *Add archetype* button to add an archetype reference (in this case Body Weight archetype, see Figure 4). Double click on the archetype or just select it and click on the *Accept* button. Once the archetype reference is added, we can select an element from it. Archetype references are stored in the definition section of the guideline and can be reused to select several elements from them (see Figure 5).




Figure 3: Select element instance




Figure 4: Adding body weigh archetype reference




Figure 5: Selecting an element from the archetype

After selecting the element, we will be presented with a new dialog for selecting the attribute of the element (see Figure 6). We choose the units attribute and click *Accept*.




Figure 6: Select attribute

Now we can select the operator and the units (*kg*) we want to specify.




Figure 7: Specifying kg units for body weight

We can do the same procedure for the archetype *Height/Length*, using the *cm* units. Next we will need to specify that we are interested at calculating the BMI from only the most recent measurements of height and weight. To do so, we need go the definition tab (Figure 8). Double click on the *Predicate (Function)* and drag it inside the *body_weight* archetype instantiation. Once it is added, click on the *element* and select *Event time*. After selecting the element we set it at max so that it will always select the most recent measurement of weight. The same process can be repeated for the *Height/Length* archetype.




Figure 8: Predicate (Function) for Weight

Next we will need to add an action that will update the magnitude attribute of the element Body Mass Index archetype. We double click on the *Action Set (Attribute)*, located at the lower panel. We follow the same steps as seem before for selecting units, but instead select magnitude attribute. By now, the rule should look like Figure 9.




Figure 9: BMI magnitude selected

Last part requires defining the expression for calculating the BMI. We click on the *Expression* link to open the expression editor and enter the equation $(weight / ((height / 100) ^ 2))$. Click on *Accept* button to add the expression to the action.

Expressions on current version must follow the pattern '(expression operator expression)'




Figure 10: Expression for BMI calculation

We can now repeat the same task to set units and precision of the element calculated.




Figure 11: BMI calculation actions

Now the guide is ready for testing. We use the *Generate form* action to see if the rule is behaving properly.

Figure 12: Form generator calculating BMI

The final GDL file will look like can be found at the Appendix A.

CHA2DS2-VASC CALCULATION

CHA2DS2-VASc is a clinical prediction score for estimating risk of stroke in patients with non –rheumatic atrial fibrillation (AF). We will see a small example on how to create a guideline in GDL to perform this calculation automatically.

(b) Risk factor-based approach expressed as a point based scoring system, with the acronym CHA ₂ DS ₂ -VASc (Note: maximum score is 9 since age may contribute 0, 1, or 2 points)	
Risk factor	Score
Congestive heart failure/LV dysfunction	1
Hypertension	1
Age ≥75	2
Diabetes mellitus	1
Stroke/TIA/thrombo-embolism	2
Vascular disease ^a	1
Age 65–74	1
Sex category (i.e. female sex)	1
Maximum score	9

Figure 13: CHA2DS2-VASC table

Before starting with the GDL editing, we must create an intermediate archetype that will store the different parts needed for the final calculation, and score summary. We create an archetype containing seven ordinals and one count (score). **The full archetype defined can be found on Appendix B.**

Once we have the archetype we can begin with the GDL editing. We create a new guide and fill up the description part. The first time the guide is executed we want to set default values to the content of the score summary archetype, so we create a rule (e.g. *Set default*), where we initialize all possible diagnosis to “Absent”. To do this we first check if all the elements have no value (using the “*Element exists*” condition).

The elements that need initializing are those referencing a diagnosis, since the age and gender should always be known, so we will only add five conditions checking for the existence of a value in the elements *Diabetes*, *Vascular diseases*, *Congestive Heart Failure*, *Previous stroke* and *Hypertension* (inside CHA2DS2-VASc archetype). The action part will consist of five actions setting these same elements to “Absent” value, using the “*Set (DataValue)*” action.

In order to avoid confusion later with the local terms we can move to the Definitions section and rename the elements to *Hypertension score*, *Previous stroke score* and so on by clicking on the “*DataValue*” of each element (Figure 14). The resulting rule should look like the one seen on Figure 15.




Figure 14: Renaming Hypertension to Hypertension score




Figure 15: Set default on CHA2DS2-VASc archetype

Now the first time the guide is executed, we have a clean instance of the archetype with the values diagnosis set to absent. Next step requires setting these diagnosis, if found on the patient's EHR.

We can start with any of the diagnosis; in this case we will use hypertension. However before we continue with this we need to create a new binding to the template of the *openEHR-EHR-EVALUATION.problem-diagnosis.v1* that will contain a subset of the ICD10 codes necessary for the diagnosis elements of CHA2DS2-VASc (for this example a template named *diagnosis_chadvas_icd10* is provided within the templates folder).

Note that we use a template in this example in order to ease the input on the generated forms.

To do so, we move to the Definitions tab and we create a new "Archetype instantiation" where we select the template (Figure 16 and Figure 17).




Figure 16: Selecting the diagnosis_chadvas_icd10 template




Figure 17: Binding of diagnosis_chadvas_icd10 template to openEHR-EHR-EVALUATION.problem-diagnosis.v1 archetype

Once we have finished with the binding to the template we can begin specifying the elements we are going to use (in this case we are going to need the diagnosis element). Next we click and drag the “*Element instantiation*” inside the *openEHR-EHR-EVALUATION.problem-diagnosis.v1* instantiation. Once it is added click on the element link, select *Diagnosis* and replace the name of *Diagnosis* with *Hypertension diagnosis*. This will be used later on the rule for detecting hypertension. The definition should look so far as in Figure 18.




Figure 18: Instantiating Diagnosis as Hypertension diagnosis

The next step is to specify that we are only interested on the instances containing hypertension problems.

To filter all other problems we will use “*Predicate (Data Value)*”. Click and drag the “*Predicate (Data Value)*” inside the *openEHR-EHR-EVALUATION.problem-diagnosis.v1* instantiation. We click on the element and select *Diagnosis* again using an “*is_a*” condition. Next we click on the “*DataValue*” and a new window will open where we click “*Select term*”. This will open another window where we will click on the button “*Add local term*” in order to create a local code of Hypertension (Figure 19) which we will bind to an ICD10 external terminology on the next step.




Figure 19: Local code for Hypertension

So far the definition should look as in Figure 20. The same steps can be repeated for the other diagnosis parts.




Figure 20: Definition of Hypertension diagnosis

The next step requires binding the diagnosis of Hypertension to an external ICD10 terminology code. To do so, we click on the “Add binding button” and we select ICD10. In order to select the local code for Hypertension we click on the + button and then on select, where we are able to choose Hypertension.




Figure 21: Selecting the local code of Hypertension for terminology binding

Next we double click on the magnifying glass under Terminology codes and a new window will open where we can select the appropriate external terminologies (for this example we use the following codes: I10, I11, I12, I13, I15).




Figure 22: Selecting the external ICD10 terminologies for Hypertension

The diagnosis now is properly binded and filtered for Hypertension. The same steps can be repeated for the other parts (Figure 23).




Figure 23: Terminology binding of the diagnosis elements for CHA2DS2-VASc

Now we are ready to create the rules for detecting these diagnoses on the patient's EHR. We create a new rule called *Set hypertension diagnosis*; in here we will check if the patient has the diagnosis *Hypertension*, if so, we will set the value of the element *Hypertension score* in the CHA2DS2-VASc archetype instance to "Present".

To find diagnosis we create a condition using the "Element exists" where we select the *Hypertension diagnosis* element and we set it to "exists". Lastly we create an action using the "Set (DataValue)" where we select the element of *Hypertension score* and we set it to "Present".



Figure 24: Set Hypertension rule

Once we have finished creating the same rules for the other diagnosis parts of CHA2DS2-VASc we can start creating the remaining rules for the gender and the age.

Before we begin editing the rules we switch to the definitions tab where we create a new “Archetype instantiation” of the *openEHR-EHR-OBSERVATION.basic_demographic.v1* archetype. Next we click on “Element instantiation” and we drag it inside the archetype. Finally, we click on the element link and we select *Gender*. We can repeat this step for *Birthdate* as well.

Since we are interested at the latest update of Gender and Birthdate, similar as before we create a condition using “Predicate (Function)” where we select the element *Event time* and we set it to *max*.




Figure 25: Predicate (Function) for Gender and Age

Now we are ready to start editing the rules for Gender and Age.

In order to set the gender of the patient we create a new rule called *Set gender female*. In here we will check if the patient's gender is female and if so to set the value of the element gender in the CHA2DS2-VASc archetype to "Female". To do so, we create a condition using "Compare (DataValue)" where we select as element *Gender* and as "DataValue" *Female*.

Next we create a new action using the "Set (DataValue)" where we select the element *Gender* from the CHA2DS2-VASc archetype and *Female* as "DataValue". The resulting rule should look as in Figure 26.

We can repeat the same process for the male gender.




Figure 26: Set Gender rule

The next step is to calculate the age of the patient.

To do this we create a new rule called *Set age above or equals to 75*. In there we create a new condition using the "Compare (Attribute)". We click on the element attribute link and we select "Birthdate.value".

Next we set the operator to "<=" and we click on the Expression link. This will open the Expression editor where we can enter the equation $(Current\ Date/Time - 75, a)$ which will calculate the age of the patient (Figure 27).

After setting the expression, we can create a new action using "Set (DataValue)". We select the element *Age* from CHA2DS2-VASc and we set it to "Above or equals to 75". By now the rule should look like Figure 28.

The same process can be repeated for the ages below 65 and between 65 and 74 (Figure 29 and Figure 30).




Figure 27: Expression for Age calculation




Figure 28: Set age above or equals to 75 rule




Figure 29: Set age between 65-74 rule




Figure 30: Set age below 65

The last part requires defining the expression for calculating the CHA2DS2-VASc score.

We create a new rule called *Calculate total score*. In there we create a new action using the "Set (Attribute)" and we click on the "Element attribute" link. This will open a new window where we click on "Add element" button. On the new window we select the CHA2DS2-VASc archetype. From there we select the "Total score.magnitude".




Figure 31: Selecting the openEHR-EHR-OBSERVATION.chadsvas_score.v1 archetype




Figure 32: Total score element

Finally we open the expression editor where we add all the elements that are needed for CHA2DS2-VASc (Figure 33).




Figure 33: Expression for calculation of CHA2DS2-VASc score

Once we have finished setting up all the rules we can now set a precondition so the guideline runs only for patients affected by non-rheumatic atrial fibrillation.

To do so we move to the definition section and we repeat the same steps as with the Hypertension diagnosis (for the *Atrial fibrillation* binding we use the ICD10 code I48). The definition should look so far as in Figure 34.




Figure 34: Definition of Atrial fibrillation diagnosis

Finally we move to the *Preconditions* section and we create a new condition using “*Element exists*” which will assess the diagnosis of *Atrial fibrillation*.




Figure 35: Atrial fibrillation precondition

The guide is now ready for testing. We can click on Generate form button and see if the rules behave properly.

The screenshot shows the 'Execution of CHA2DS2-VASc score calculation' window. The 'Input' section contains the following data:

- openEHR-EHR-EVALUATION.problem-diagnosis.v1: EHR cT Diagnosis = cT Atrial fibrillation and flutter
- openEHR-EHR-EVALUATION.problem-diagnosis.v1: EHR cT Diagnosis = cT Hypertensive renal disease
- openEHR-EHR-OBSERVATION.basic_demographic.v1: EHR cT Gender = Female
- EHR Birthdate = 08/06/1986 12:31:52
- EHR Event time = 04/12/2013 12:32:11

The 'Result' section shows the calculated scores:

- openEHR-EHR-OBSERVATION.chadsvas_score.v1: CDS Congestive Heart Failure score = 0 - Absent
- CDS Hypertension score = 1 - Present
- CDS Diabetes score = 0 - Absent
- CDS Age = 0 - Under 65
- CDS Previous stroke score = 0 - Absent
- CDS Vascular diseases score = 0 - Absent
- CDS Gender = 1 - Female
- CDS Total score = 2

Figure 36: Execution of CHA2DS2-VASc score calculation

BMI GDL

```

(GUIDE) <
  gdl_version = <"0.1">
  id = <"BMI.Calculation.v.1">
  concept = <"gt0000">
  language = (LANGUAGE) <
    original_language = <[ISO_639-1::en]>
  >
  description = (RESOURCE_DESCRIPTION) <
    details = <
      ["en"] = (RESOURCE_DESCRIPTION_ITEM) <
        >
      >
    lifecycle_state = <"Author draft">
  >
  definition = (GUIDE_DEFINITION) <
    archetype_bindings = <
      [1] = (ARCHETYPE_BINDING) <
        archetype_id = <"openEHR-EHR-OBSERVATION.body_weight.v1">
        domain = <"EHR">
        elements = <
          ["gt0002"] = (ELEMENT_BINDING) <
            path = <"/data[at0002]/events[at0003]/data[at0001]/items[at0004]">
          >
        >
        predicates = <"max(/event/time)",...>
      >
      [2] = (ARCHETYPE_BINDING) <
        archetype_id = <"openEHR-EHR-OBSERVATION.height.v1">
        domain = <"EHR">
        elements = <
          ["gt0003"] = (ELEMENT_BINDING) <
            path = <"/data[at0001]/events[at0002]/data[at0003]/items[at0004]">
          >
        >
        predicates = <"max(/event/time)",...>
      >
      [3] = (ARCHETYPE_BINDING) <
        archetype_id = <"openEHR-EHR-OBSERVATION.body_mass_index.v1">
        domain = <"CDS">
        elements = <
          ["gt0004"] = (ELEMENT_BINDING) <
            path = <"/data[at0001]/events[at0002]/data[at0003]/items[at0004]">
          >
        >
      >
    >
  >
  rules = <
    ["gt0001"] = (RULE) <
      when = <"$gt0002.units=='kg'", "$gt0003.units=='cm'">
      then = <"$gt0004.precision=2", "$gt0004.units='kg/m2'",
"$gt0004.magnitude=($gt0002.magnitude/(( $gt0003.magnitude/100)^2))">
      priority = <1>
    >
  >
  >
  ontology = (GUIDE_ONTOLOGY) <
    term_definitions = <
      ["en"] = (TERM_DEFINITION) <
        terms = <
          ["gt0000"] = (TERM) <
            text = <"Body Mass Index calculation">

```

```

    >
    ["gt0001"] = (TERM) <
      text = <"Calculate body mass index">
    >
    ["gt0002"] = (TERM) <
      text = <"Weight">
      description = <"The weight of the individual.">
    >
    ["gt0003"] = (TERM) <
      text = <"Height/Length">
      description = <"The length of the body from crown of head to sole of foot.">
    >
    ["gt0004"] = (TERM) <
      text = <"Body Mass Index">
      description = <"Index describing ratio of weight to height.">
    >
  >
>
>
>
>
>

```

CHA2DS2-VASC CALCULATION GDL

```

(GUIDE) <
  gdl_version = <"0.1">
  id = <"CHA2DS2VASC_Score_calculation.v1">
  concept = <"gt0036">
  language = (LANGUAGE) <
    original_language = <[ISO_639-1::en]>
  >
  description = (RESOURCE_DESCRIPTION) <
    details = <
      ["en"] = (RESOURCE_DESCRIPTION_ITEM) <
        copyright = <" ">
        keywords = <"Atrial Fibrillation", "Stroke", "CHA2DS2-VASC">
        misuse = <" ">
        purpose = <"Calculates stroke risk for patients with atrial fibrillation, possibly better than the CHADS2 score.">
        use = <"Calculates stroke risk for patients with atrial fibrillation, possibly better than the CHADS2 score.">
      >
      ["sv"] = (RESOURCE_DESCRIPTION_ITEM) <
      >
    >
  lifecycle_state = <"Author draft">
  original_author = <
    ["date"] = <"2012/12/03">
    ["email"] = <"rong.chen@cambio.se">
    ["name"] = <"Rong Chen">
    ["organisation"] = <"Cambio Healthcare Systems">
  >
  other_contributors = <"Carlos Valladares",...>
  other_details = <
    ["references"] = <"1. Lip GY, Nieuwlaat R, Pisters R, Lane DA, Crijns HJ. Refining clinical risk stratification for predicting stroke and thromboembolism in atrial fibrillation using a novel risk factor-based approach: the euro heart survey on atrial fibrillation. Chest. 2010 Feb;137(2):263-72. Epub 2009 Sep 17. PubMed PMID: 19762550.
2. European Heart Rhythm Association; European Association for Cardio-Thoracic Surgery, Camm AJ, Kirchhof P, Lip GY, Schotten U, Savelieva I, Ernst S, Van Gelder IC, Al-Attar N, Hindricks G, Prendergast B, Heidbuchel H, Alfieri O, Angelini A, Atar D, Colonna P, De Caterina R, De Sutter J, Goette A, Gorenek B, Heldal M, Hohloser SH, Kolh P, Le Heuzey JY, Ponikowski P, Rutten FH. Guidelines for the management of atrial fibrillation: the Task Force for the Management of Atrial Fibrillation of the European Society of Cardiology (ESC). Eur Heart J. 2010 Oct;31(19):2369-429. Epub 2010 Aug 29. Erratum in: Eur Heart J. 2011 May;32(9):1172. PubMed PMID: 20802247.
3. Lip GY, Frison L, Halperin JL, Lane DA. Identifying patients at high risk for stroke despite anticoagulation: a comparison of contemporary stroke risk stratification schemes in an anticoagulated atrial fibrillation cohort. Stroke. 2010 Dec;41(12):2731-8. Epub 2010 Oct 21. PubMed PMID: 20966417.">
    >
  >
  definition = (GUIDE_DEFINITION) <

```

```

archetype_bindings = <
  [1] = (ARCHETYPE_BINDING) <
    archetype_id = <"openEHR-EHR-EVALUATION.problem-diagnosis.v1">
    domain = <"EHR">
    elements = <
      ["gt0107"] = (ELEMENT_BINDING) <
        path = <"/data[at0001]/items[at0002.1]">
      >
    >
    predicates = <"/data[at0001]/items[at0002.1] is_a local::gt0105|Atrial fibrillation|",...>

    template_id = <"diagnosis_chadvas_icd10">

  >

  [2] = (ARCHETYPE_BINDING) <
    archetype_id = <"openEHR-EHR-EVALUATION.problem-diagnosis.v1">
    domain = <"EHR">
    elements = <
      ["gt0108"] = (ELEMENT_BINDING) <
        path = <"/data[at0001]/items[at0002.1]">
      >
    >
    predicates = <"/data[at0001]/items[at0002.1] is_a local::gt0101|Hypertension|",...>

    template_id = <"diagnosis_chadvas_icd10">

  >

  [3] = (ARCHETYPE_BINDING) <
    archetype_id = <"openEHR-EHR-EVALUATION.problem-diagnosis.v1">
    domain = <"EHR">
    elements = <
      ["gt0109"] = (ELEMENT_BINDING) <
        path = <"/data[at0001]/items[at0002.1]">
      >
    >
    predicates = <"/data[at0001]/items[at0002.1] is_a local::gt0102|Diabetes|",...>

    template_id = <"diagnosis_chadvas_icd10">

  >

  [4] = (ARCHETYPE_BINDING) <
    archetype_id = <"openEHR-EHR-EVALUATION.problem-diagnosis.v1">
    domain = <"EHR">
    elements = <
      ["gt0110"] = (ELEMENT_BINDING) <
        path = <"/data[at0001]/items[at0002.1]">
      >
    >
    predicates = <"/data[at0001]/items[at0002.1] is_a local::gt0100|Congestive heart
failure|",...>

    template_id = <"diagnosis_chadvas_icd10">

  >

  [5] = (ARCHETYPE_BINDING) <
    archetype_id = <"openEHR-EHR-EVALUATION.problem-diagnosis.v1">
    domain = <"EHR">
    elements = <
      ["gt0111"] = (ELEMENT_BINDING) <
        path = <"/data[at0001]/items[at0002.1]">
      >
    >
    predicates = <"/data[at0001]/items[at0002.1] is_a local::gt0103|Previous stroke or TIA|",...>

    template_id = <"diagnosis_chadvas_icd10">

  >

  [6] = (ARCHETYPE_BINDING) <

```



```

archetype_id = <"openEHR-EHR-EVALUATION.problem-diagnosis.v1">
domain = <"EHR">
elements = <
  ["gt0112"] = (ELEMENT_BINDING) <
    path = <"/data[at0001]/items[at0002.1]">
  >
>
predicates = <"/data[at0001]/items[at0002.1] is_a local::gt0104|Vascular disease|",...>

template_id = <"diagnosis_chadvas_icd10">

>

[7] = (ARCHETYPE_BINDING) <
archetype_id = <"openEHR-EHR-OBSERVATION.basic_demographic.v1">
domain = <"EHR">
elements = <
  ["gt0006"] = (ELEMENT_BINDING) <
    path = <"/data[at0001]/events[at0002]/data[at0003]/items[at0004]">
  >
  ["gt0113"] = (ELEMENT_BINDING) <
    path = <"/data[at0001]/events[at0002]/data[at0003]/items[at0008]">
  >
>
predicates = <"max(/event/time)",...>
>

[8] = (ARCHETYPE_BINDING) <
archetype_id = <"openEHR-EHR-OBSERVATION.chadvas_score.v1">
domain = <"CDS">
elements = <
  ["gt0009"] = (ELEMENT_BINDING) <
    path = <"/data[at0002]/events[at0003]/data[at0001]/items[at0046]">
  >
  ["gt0010"] = (ELEMENT_BINDING) <
    path = <"/data[at0002]/events[at0003]/data[at0001]/items[at0032]">
  >
  ["gt0011"] = (ELEMENT_BINDING) <
    path = <"/data[at0002]/events[at0003]/data[at0001]/items[at0035]">
  >
  ["gt0012"] = (ELEMENT_BINDING) <
    path = <"/data[at0002]/events[at0003]/data[at0001]/items[at0026]">
  >
  ["gt0013"] = (ELEMENT_BINDING) <
    path = <"/data[at0002]/events[at0003]/data[at0001]/items[at0039]">
  >
  ["gt0014"] = (ELEMENT_BINDING) <
    path = <"/data[at0002]/events[at0003]/data[at0001]/items[at0029]">
  >
  ["gt0015"] = (ELEMENT_BINDING) <
    path = <"/data[at0002]/events[at0003]/data[at0001]/items[at0042]">
  >
  ["gt0016"] = (ELEMENT_BINDING) <
    path = <"/data[at0002]/events[at0003]/data[at0001]/items[at0099]">
  >
>
>
>
pre_conditions = <"$gt0107!=null",...>
rules = <
  ["gt0017"] = (RULE) <
    when = <"$gt0110!=null",...>
    then = <"$gt0012=1|local::at0028|Present|",...>
    priority = <10>
  >
  ["gt0018"] = (RULE) <
    when = <"$gt0108!=null",...>
    then = <"$gt0014=1|local::at0031|Present|",...>
    priority = <11>
  >
  ["gt0019"] = (RULE) <
    when = <"$gt0109!=null",...>
    then = <"$gt0010=1|local::at0034|Present|",...>
    priority = <9>

```



```

>
["gt0020"] = (RULE) <
  when = <"$gt0111!=null",...>
  then = <"$gt0013=2|local::at0041|Present|",...>
  priority = <8>
>
["gt0021"] = (RULE) <
  when = <"$gt0113.value>($currentDateTime.value-65,a)",...>
  then = <"$gt0011=0|local::at0036|Under 65|",...>
  priority = <3>
>
["gt0022"] = (RULE) <
  when = <"$gt0113.value<=($currentDateTime.value-65,a)",
"$gt0113.value>($currentDateTime.value-75,a)">
  then = <"$gt0011=1|local::at0037|Between 65-74|",...>
  priority = <2>
>
["gt0023"] = (RULE) <
  when = <"$gt0113.value<=($currentDateTime.value-75,a)",...>
  then = <"$gt0011=2|local::at0038|Above or equals to 75|",...>
  priority = <4>
>
["gt0024"] = (RULE) <
  when = <"$gt0006==local::at0005|Male|",...>
  then = <"$gt0015=0|local::at0043|Male|",...>
  priority = <5>
>
["gt0025"] = (RULE) <
  when = <"$gt0006==local::at0006|Female|",...>
  then = <"$gt0015=1|local::at0044|Female|",...>
  priority = <6>
>
["gt0026"] = (RULE) <
  then =
<"$gt0016.magnitude((((($gt0009.value+$gt0010.value)+$gt0011.value)+$gt0015.value)+$gt0012.value)+$g
t0013.value)+$gt0014.value)",...>
  priority = <1>
>
["gt0027"] = (RULE) <
  when = <"$gt0010==null", "$gt0009==null", "$gt0012==null", "$gt0013==null", "$gt0014==null">
  then = <"$gt0010=0|local::at0033|Absent|", "$gt0009=0|local::at0047|Absent|",
"$gt0012=0|local::at0027|Absent|", "$gt0013=0|local::at0040|Absent|",
"$gt0014=0|local::at0030|Absent|">
  priority = <12>
>
["gt0028"] = (RULE) <
  when = <"$gt0112!=null",...>
  then = <"$gt0009=1|local::at0048|Present|",...>
  priority = <7>
>
>
>
ontology = (GUIDE_ONTOLOGY) <
  term_bindings = <
    ["ICD10"] = (TERM_BINDING) <
      bindings = <
        ["gt0100"] = (BINDING) <
          codes = <[ICD10::I50],...>
          uri = <"">
        >
        ["gt0101"] = (BINDING) <
          codes = <[ICD10::I10], [ICD10::I11], [ICD10::I12], [ICD10::I13], [ICD10::I15]>
          uri = <"">
        >
      >
    >
    ["gt0102"] = (BINDING) <
      codes = <[ICD10::E10], [ICD10::E11], [ICD10::E12], [ICD10::E13], [ICD10::E14]>
      uri = <"">
    >
  >

```

```

>

["gt0103"] = (BINDING) <

  codes = <[ICD10::I63], [ICD10::I693], [ICD10::G459]>

  uri = <"">

>

["gt0104"] = (BINDING) <

  codes = <[ICD10::I21], [ICD10::I22], [ICD10::I249], [ICD10::I250], [ICD10::I251],
[ICD10::I252], [ICD10::I255], [ICD10::I256], [ICD10::I258], [ICD10::I259], [ICD10::Z951],
[ICD10::I739], [ICD10::I70], [ICD10::I71], [ICD10::I72]>

  uri = <"">

>

["gt0105"] = (BINDING) <

  codes = <[ICD10::I48],...>

  uri = <"">

>
>
>
>
term_definitions = <
  ["en"] = (TERM_DEFINITION) <
    terms = <
      ["gt0003"] = (TERM) <
        text = <"Diagnosis">
        description = <"The index diagnosis">
      >
      ["gt0005"] = (TERM) <
        text = <"Age">
        description = <"*">
      >
      ["gt0006"] = (TERM) <
        text = <"Gender">
        description = <"*">
      >
      ["gt0009"] = (TERM) <
        text = <"Vascular diseases">
        description = <"Prior myocardial infarction, Peripheral artery disease, Aortic plaque">
      >
      ["gt0010"] = (TERM) <
        text = <"Diabetes">
        description = <"*">
      >
      ["gt0011"] = (TERM) <
        text = <"Age">
        description = <"*">
      >
      ["gt0012"] = (TERM) <
        text = <"Congestive Heart Failure">
        description = <"*">
      >
      ["gt0013"] = (TERM) <
        text = <"Previous stroke">
        description = <"*">
      >
      ["gt0014"] = (TERM) <
        text = <"Hypertension">
        description = <"*">
      >
      ["gt0015"] = (TERM) <
        text = <"Gender">

```

```

    description = <"*">
  >
  ["gt0016"] = (TERM) <
    text = <"Total score">
    description = <"*">
  >
  ["gt0017"] = (TERM) <
    text = <"Set CHF present">
  >
  ["gt0018"] = (TERM) <
    text = <"Set hypertension">
  >
  ["gt0019"] = (TERM) <
    text = <"Set diabetes">
  >
  ["gt0020"] = (TERM) <
    text = <"Set previous stroke">
  >
  ["gt0021"] = (TERM) <
    text = <"Set age below 65">
  >
  ["gt0022"] = (TERM) <
    text = <"Set age between 65-74">
  >
  ["gt0023"] = (TERM) <
    text = <"Set age above or equals to 75">
  >
  ["gt0024"] = (TERM) <
    text = <"Set gender male">
  >
  ["gt0025"] = (TERM) <
    text = <"Set gender female">
  >
  ["gt0026"] = (TERM) <
    text = <"Calculate total score">
  >
  ["gt0027"] = (TERM) <
    text = <"Set default">
  >
  ["gt0028"] = (TERM) <
    text = <"Set vascular disease">
  >
  ["gt0036"] = (TERM) <
    text = <"CHA2DS2-VASc Score">
    description = <"CHA2DS2-VASc Score for estimation stroke risks in atrial fibrillation">
  >
  ["gt0100"] = (TERM) <
    text = <"Congestive heart failure">
  >
  ["gt0101"] = (TERM) <
    text = <"Hypertension">
  >
  ["gt0102"] = (TERM) <
    text = <"Diabetes">
  >
  ["gt0103"] = (TERM) <
    text = <"Previous stroke or TIA">
  >
  ["gt0104"] = (TERM) <
    text = <"Vascular disease">
  >
  ["gt0105"] = (TERM) <
    text = <"Atrial fibrillation">
  >
  ["gt0107"] = (TERM) <
    text = <"Atrial fibrillation diagnosis">
  >
  ["gt0108"] = (TERM) <
    text = <"Hypertension diagnosis">
  >
  ["gt0109"] = (TERM) <
    text = <"Diabetes diagnosis">
  >
  ["gt0110"] = (TERM) <
    text = <"Congestive heart failure diagnosis">

```

```

    >
    ["gt0111"] = (TERM) <
      text = <"Previous stroke or TIA diagnosis">
    >
    ["gt0112"] = (TERM) <
      text = <"Vascular disease diagnosis">
    >
    ["gt0113"] = (TERM) <
      text = <"Birthdate">
      description = <"*">
    >
  >
  >
  ["sv"] = (TERM_DEFINITION) <
    terms = <
      ["gt0003"] = (TERM) <
        text = <"Diagnos">
        description = <"*The index diagnosis (en)">
      >
      ["gt0005"] = (TERM) <
        text = <"Ålder">
        description = <"** (en)">
      >
      ["gt0006"] = (TERM) <
        text = <"Kön">

        description = <"** (en)">
      >

      ["gt0009"] = (TERM) <

        text = <"Vaskulär sjukdom">

        description = <"*Prior myocardial infarction, Peripheral artery disease, Aortic plaque
(en)">
      >
      ["gt0010"] = (TERM) <

        text = <"Diabetes">

        description = <"** (en)">
      >

      ["gt0011"] = (TERM) <

        text = <"Ålder">

        description = <"** (en)">
      >

      ["gt0012"] = (TERM) <

        text = <"Hjärtsvikt/VK-dysfunktion">

        description = <"** (en)">
      >
      ["gt0013"] = (TERM) <
        text = <"Stroke/TIA">
        description = <"** (en)">
      >
      ["gt0014"] = (TERM) <
        text = <"Hypertoni">
        description = <"** (en)">
      >
      ["gt0015"] = (TERM) <
        text = <"Kön">
        description = <"** (en)">
      >
      ["gt0016"] = (TERM) <

```

```

    text = <"Poäng">
    description = <"** (en)">
  >
  ["gt0017"] = (TERM) <
    text = <"*Set CHF present (en)">
  >
  ["gt0018"] = (TERM) <
    text = <"*Set hypertension (en)">
  >
  ["gt0019"] = (TERM) <
    text = <"*Set diabetes (en)">
  >
  ["gt0020"] = (TERM) <
    text = <"*Set previous stroke (en)">
  >
  ["gt0021"] = (TERM) <
    text = <"*Set aget below 65 (en)">
  >
  ["gt0022"] = (TERM) <
    text = <"*Set age between 65-74 (en)">
  >
  ["gt0023"] = (TERM) <
    text = <"*Set age above or equals to 75 (en)">
  >
  ["gt0024"] = (TERM) <
    text = <"*Set gender male (en)">
  >
  ["gt0025"] = (TERM) <
    text = <"*Set gender female (en)">
  >
  ["gt0026"] = (TERM) <
    text = <"*Calculate total score (en)">
  >
  ["gt0027"] = (TERM) <
    text = <"*Set default (en)">
  >
  ["gt0028"] = (TERM) <
    text = <"*Set vascular disease (en)">
  >
  ["gt0036"] = (TERM) <
    text = <"CHA2DS2-VASc Poäng">
    description = <"*CHA2DS2-VASc Score for estimation stroke risks in atrial fibrillation
(en)">
  >

  ["gt0100"] = (TERM) <

    text = <"Hjärtsvikt/VK-dysfunktion">

  >
  ["gt0101"] = (TERM) <
    text = <"Hypertoni">
  >
  ["gt0102"] = (TERM) <
    text = <"Diabetes">
  >
  ["gt0103"] = (TERM) <
    text = <"Stroke/TIA">
  >

  ["gt0104"] = (TERM) <

    text = <"Vaskulär sjukdom">

  >

  ["gt0105"] = (TERM) <

    text = <"* Atrial fibrillation (en)">

```

```
>

["gt0107"] = (TERM) <

  text = <"*Atrial fibrillation diagnosis (en)">

>

["gt0108"] = (TERM) <

  text = <"*Hypertension diagnosis (en)">

>

["gt0109"] = (TERM) <

  text = <"*Diabetes diagnosis (en)">

>

["gt0110"] = (TERM) <

  text = <"*Congestive heart failure diagnosis (en)">
>
["gt0111"] = (TERM) <
  text = <"*Previous stroke or TIA diagnosis (en)">
>
["gt0112"] = (TERM) <
  text = <"*Vascular disease diagnosis (en)">
>
["gt0113"] = (TERM) <
>
>
>
>
>
>
```

APPENDIX B – ARCHETYPES

CHA2DS2-VASC ARCHETYPE

Arketypeditor [en] CHADSVAS Score

Fil Edit Språk Terminologi Visa Tools Hjälp

openEHR-EHR-OBSERVATION.chadsvas_score.v1

Huvud Definition Terminologi Visa Gränssnitt Beskrivning

☐ *Protocol (en) ☐ Participation ☐ *Person State with History (en)

Data

☐ *Person State (en)

Träd Händelser

Struktur Träd

☐ Ordnad

Kardinalitet

Min: 1 Max: 1 ☐ *Unbounded (en)

at0026

- + Congestive Heart Failure
- Hypertension
- Diabetes
- ↑ Age
- ↓ Previous stroke
- Vascular diseases
- Gender
- 1,2,3 Total score

Restriktioner Details

Förekomster

Min: 0 Max: 1 ☐ *Unbounded (en)

Beskrivning *

Runtime namn restriktion

Ordinal

	C	Text	Description
▶	0	Absent	*
	1	Present	*
*			

☐ Antaget värde